
Wprowadzenie

Charakterystyka robota MOTOMAN UP6

Robot  MOTOMAN  UP6  jest  to  sześcioosiowy  robot  przegubowy  o  kinematyce
PUMA. Robot ten ze względu na swoją funkcjonalność i dokładność wykorzystywany jest
przede  wszystkim w procesie  spawania.  Dlatego  standardowo w sterowniku  znajdują  się
funkcje obsługujące urządzenia spawalnicze.

Rys. 1. Robot przemysłowy YASKAWA MOTOMAN UP6

Napędy  robota  stanowią  zespół  sześciu  silników  prądu  zmiennego  3-fazowego
wyposażonych w absolutne przetworniki obrotowo-impulsowe sterowane przez serwonapędy.
Kombinacja  poszczególnych  osi  wraz  z  określonymi  wymiarami  członów  robota  daje
dopuszczalny kształt i rozmiary przestrzeni roboczej.



Rys. 2. Przestrzeń operacyjna (robocza) robota przemysłowego YASKAWA
MOTOMAN UP6

Sterowanie robotem odbywa się za pomocą panelu operatorskiego zamieszczonego na
szafie  sterowniczej.  Zawiera  on  główny  wyłącznik  zasilania,  przycisk  stopu  awaryjnego,
przycisk informujący o stanie pracy serwonapędu, lampkę oznaczającą wystąpienie alarmu,
przycisk  sygnalizujący  bezwarunkowe  zatrzymanie  programu,  przycisk  rozpoczęcia
wykonywania  programu  oraz  przyciski  oznaczające  pracę  w  określonym  trybie:  trybie
programowania, trybie pracy automatycznej poprzez urządzenia zewnętrzne, trybie pracy.

Robot  wyposażony jest  także w panel  programisty,  który pełni  rolę  monitora  oraz
klawiatury  jednocześnie.  Przy  pomocy  tego  panelu  operator  może  wykonać  większość
czynności związanych z obsługą i programowaniem robota.



Rys. 3. Panel operatorski robota przemysłowego YASKAWA MOTOMAN UP6

Głównymi klawiszami jakimi należy się posługiwać podczas ćwiczenia są:

Czterokierunkowy klawisz nawigacyjny służący do poruszania się po menu oraz 
podczas naciśnięcia razem z przyciskiem SHIFT, do przewijania ekranu.

Klawisz wyboru funkcji.

Klawisz szybkiego wyboru menu głównego. Z klawiszem SHIFT powoduje 
przejście na wyższy poziom menu.

Klawisz przełączania pomiędzy menu głównym a funkcjami ekranowymi. Z 
klawiszem SHIFT przełącza aktywne języki menu.

Klawisz przewijania kolejnych ekranów na danym poziomie menu w lewo. Z 
wciśniętym klawiszem SHIFT przewija w prawo.



Klawisz blokady trybu uczenia.

Lampka kontrolna stanu serwonapędu.

Zestaw 12 przycisków służących do sterowania poszczególnymi 
osiami robota w zależności od wyboru układu współrzędnych.

Przyciski służące do cyklicznego wyboru zakresu prędkości robota.

Przycisk szybkiego ruchu robota.

Przycisk wyboru układu współrzędnych aktywnego podczas pracy robota.

Aktywność tego przycisku (świecąca dioda) informuje nas, że klawisze sterowania 
ruchem osi robota odnoszą się do podstawowego zestawu osi, czyli każda para 
klawiszy steruje jedną osią.

Klawiatura numeryczna.

Przycisk umożliwiający wstawianie instrukcji do programu.

Przycisk umożliwiający modyfikację danych liczbowych, funkcji lub jej opcji.

Usuwanie aktywnej wartości lub instrukcji.

Wprowadzenie aktualnej wartości z bufora edycyjnego do instrukcji.



Zaprzestanie wprowadzania zmian.

Wyświetlenie listy dostępnych funkcji w trakcie edycji programu.

Zmiana działania innych klawiszy.

Przy równoczesnym naciśnięciu INTER LOCK uruchamia program sterujący w 
jednym z wybranych trybów pracy.

Przy równoczesnym naciśnięciu INTER LOCK uruchamia kolejne bloki programu 
sterującego zgodnie z ich kolejnością.

Przy równoczesnym naciśnięciu INTER LOCK uruchamia kolejne bloki programu 
sterującego w odwrotnej kolejności.

Poza  opisanymi  przyciskami  możemy  wyróżnić  również  przycisk  blokady  trybu
uczenia, który znajduje się z tyłu panelu operatorskiego. Jest to przełącznik trójpołożeniowy,
który jest aktywny jeśli znajduje się w położeniu środkowym. Pełne dociśnięcie lub jego brak
powoduje,  że  panel  jest  nieaktywny.  Stanowi  to  zabezpieczenie  m.  in.  przed  porażeniem
prądem.

Na wyświetlaczy panelu operatorskiego widoczne jest menu główne podzielone na 8 
podstawowych pozycji:

 JOB – zarządzanie programami, wybór aktywnego programu, tworzenie i kasowanie 
programów oraz ich modyfikacja.

 ARC WELDING – ustawienia dotyczące procesu spawania.
 VARIABLE – edycja zmiennych oraz współrzędnych punktów używanych w 

programach sterujących.
 IN/OUT – obsługa sygnałów wejść/wyjść dyskretnych robota.
 ROBOT – ustawienia dotyczące układów współrzędnych użytkownika, danych 

narzędziowych, pozycji osi sterowanych numerycznie, punktów referencyjnych itp.
 SYSTEM INFO – diagnostyka systemu, historia komunikatów i alarmów.
 FD/PC CARD – wykonywanie kopii systemu, programów, innych danych oraz ich 

odtwarzanie poprzez złącza komunikacyjne.
 SETUP – ogólne ustawienia systemu istotne z punktu widzenia programowania, 

wykonywania programów oraz systemów bezpieczeństwa.



Przebieg ćwiczenia

Etapy tworzenia programu.

1.) Wybranie na sterowniku trybu TEACH.
2.) Zapoznanie się z obsługą teach pendant'a.
3.) Ustawienie poziomu czułości czujnika kolizji na wartość równą 8 – dla trybu uczenia 
robota . W przypadku pracy normalnej   na 9.
4.) Zdefiniowanie układu współrzędnych użytkownika- kartezjański prawoskrętny układ 
współrzędnych.
5.) Zapisanie pozycji charakterystycznych.

W celu utworzenie oraz modyfikowania nowego programu sterującego należy:
 Z menu głównego wybrać funkcję JOB
 Wybrać opcję CREATE NEW JOB
 Wprowadzić nazwę nowego programu w polu JOB NAME
 Wprowadzić komentarz do programu w polu COMMENT
 Wybrać i zaakceptować klawiszem SELECT.

Przed  rozpoczęciem  tworzenia  programu  sterującego  należy  zdefiniować  układ
współrzędnych użytkownika w przestrzeni roboczej robota przemysłowego. Definicję układu
współrzędnych  przeprowadzić  należy  wykorzystując  pomoc  prowadzącego  ćwiczenie.  W
jednym  z  punktów  charakterystycznych  dotyczących  układu  współrzędnych  użytkownika
wybieramy punkt, który będzie punktem startowym programu, a także ustawiamy poziom
czułości czujnika zderzenia na 8. 

W następnym kroku przechodzimy do pisania programu. Zaprogramowany robot ma
za zadanie zlokalizować element,  dokonać pomiaru średnicy tego przedmiotu,  a następnie
posortować kolejne elementy w zależności od ich średnicy. 

W celu jednoznacznej identyfikacji elementu w przestrzeni roboczej wykorzystuje się
urządzenie pomiarowe współpracujące z robotem przemysłowym. Tym urządzeniem jest laser
pomiarowy  Keyence  LK-G5000.  Wykorzystując  odpowiednie  sygnały  podawane  z  lasera
pomiarowego na wejście do robota, robot dokonuje obliczeń charakterystycznych punktów
elementu mierzonego, które to następnie zapisuje do zmiennych pozycyjnych oraz zmiennych
typu DOUBLE oraz REAL. Program sterujący bazując na przechowywanych wartościach w
odpowiednich  zmiennych  –  odwołując  się  do  nich,  wykorzystuje  przechowywane  tam
określone wartości do obliczeń arytmetycznych oraz do realizacji instrukcji ruchu członów.

Aby  zrealizować  cele  ćwiczenia,  konieczne  jest  również  zapoznanie  studentów  z
charakterystyką i możliwościami lasera pomiarowego.

Ogólną charakterystykę, parametry pracy, sposób ustawienia określonych parametrów
pracy  lasera  pomiarowego  w sposób  zwięzły  przedstawiono  w opisie  znajdującym się  w
instrukcji  wykonywania  ćwiczenia.  Studenci  realizują  ustawienia  parametrów pracy lasera
pomiarowego  wyłącznie  pod  nadzorem  osoby  prowadzącej  ćwiczenia  oraz  pracownika
technicznego obsługującego laboratorium.



Ogólna charakterystyka lasera pomiarowego LK-G firmy KEYENCE

Klasyfikacja pod względem klasy laserowej LK-G przedstawia się następująco: 
Tryb montażu: dyfuzyjny
Maksymalna moc: 0.95 mW
Odległość referencyjna: 150 mm
Zakres pomiarowy: ±40 mm
Źródło światła: 

a) Typ: czerwony laser półprzewodnikowy,
b) Długość fali: 655 nm,
c) Klasa lasera: II,
d) Maksymalna moc (wyjście): 0.95 mW.

Średnica plamki (w odległości referencyjnej): ø120 μm
Liniowość: ±0.02%
Powtarzalność: 0.25 μm
Cykl  próbkowania:  2.55/5/10/20/50/100/200/500/1000  μs  (wybór  spośród  dziewięciu
kroków)
Wahania temperatury: 0.01% of F.S.
Odporność na środowisko: 

a) stopień ochrony: IP67,
b) światło otoczenia: Żarówka lub świetlówka: 5000 lux max,
c) temperatura otoczenia: 0 do +50 °C,
d) wilgotność otoczenia: 35 do 85% RH (bez kondensacji),
e) odporność na wibracje:  10 do 55 Hz, podwójna amplituda  1.5 mm, 2 godziny dla

każdej z osi XYZ.
Materiał: odlew aluminiowy
Waga: wraz z kablem ok. 300g

Rys. 4. Laser pomiarowy  LK-G firmy KEYENCE 

Producent lasera pomiarowego firma KEYENCE wyposażyła stanowisko pomiarowe
w  odpowiednie  dedykowane  oprogramowanie  LK-Nawigator  2.  



LK-Navigator  2  jest  programem do konfiguracji  parametrów i  monitorowania  sterownika
lasera  serii  LK-G5000.  Poprzez  podłączenie  komputera  i  kontrolera  lasera  używany  jest
również do konfiguracji parametrów i monitorowania stanu pracy.

 LK-Navigator 2 pozwala na przesyłanie ustawień w kontrolerze do komputera PC, w
celu przeglądania i edycji. Posiada wbudowane funkcje kopiowania i wklejania w celu
konfiguracji  wielu  sterowników i  ustawień   programów,  co  można zrealizować  w
krótkim czasie,

 LK-Navigator 2 komunikuje się z kontrolerem poprzez naciśnięcie przycisku,
 ustawienia mogą być ładowane z i zapisywane na komputerze. Ustawienia zapisane na

komputerze  można  przywrócić  ustawienia  kontrolera  w przypadku  gdy ustawienia
zostaną skasowane.

Następujące elementy mogą zostać ustawione w komputerze.

 ustawienia głowicy,
 wszystkie nastawy,
 wspólne ustawienia,
 ustawienia środowiska.

Dodatkowo jest możliwość monitorowania następujących elementów:

 wartości pomiarowej,
 fali światła odbitego.

Sterownik  posiada  funkcję  przechwytywania  (zczytywania)  danych  do
przechowywania wartości pomiarów.

Komputer klasy PC jest stosowany w celu kontrolowania następujących funkcji:

 start / stop / czyszczenie magazynu danych,
 konfiguracji metod przechowywania danych,
 wysyłania / pobierania ustawień między kontrolerem, a komputerem,
 przesyłanie danych z pamięci kontrolera,
 czytania i zapisywania danych do przechowywania na komputerze,
 wyświetlania przebiegu do 12 serii danych, pobranych ze sterownika,
 wyświetlania obliczeń na danych pomiędzy ustawieniami użytkownika.

Rys. 5. Zakładka Data Storage do zarządzania pomiarami 



Objaśnienie najważniejszych funkcji menu głównego oprogramowania LK-Navigator
2,  z  punktu  widzenia  obsługi  oprogramowania  uważa  się  za  zasadne,  ze  względu  na
wykorzystanie lasera i oprogramowania do pomiarów robotem przemysłowym MOTOMAN
UP6. Na rysunku 6 przedstawiono widok strony głównej programu LK-Navigator 2, a wykorzystując
listę numerowaną zaznaczono istotne funkcje, których obsługa i ustawienia wpływają na poprawne
prowadzenie pomiarów laserem.

Rys. 6. Widok okna głównego programu LK-Navigator 2: 1- pasek główny, 2-
pasek narzędzi, 3-panel ustawień, 4-ustawienia programu, 5-ustawienia

środowiska kontrolera, 6-przycisk pomocy, 7-zakończenie pracy z programem, 8-
wczytywanie i zapisywanie ustawień utworzonych w programie LK-Navigator 2 



Rys. 7. Oprogramowanie do lasera pomiarowego – panel z ustawieniami głowicy
pomiarowej



Rys. 8. Chwytak robota przemysłowego MOTOMAN UP6 z zamocowanym laserem
pomiarowym

Rys. 9. Sterownik lasera pomiarowego

Na  poniższym  rysunku  przedstawiono  ideę  sposobu  określania  punktów
charakterystycznych i dokonywania pomiarów na elementach, które mają być posortowane.



                                                                 

Opis istotnych instrukcji kodu źródłowego programu  sterującego:

NOP ; Początek programu

MOVL V=256 ; Ruch robota z interpolacją liniową do 
zdefiniowanego punktu 
charakterystycznego z prędkością 256 
cm/min

DOUT OT#(4) ON ; Ustawienie wartości 1 na wyjściu 
dyskretnym 4 przypisanym do chwytaka. 
Powoduje to zamknięcie chwytaka.

MOVL V=33 PL=0 UNTIL IN#(1)=ON ; Ruch powrotny z interpolacją liniową do 
początku zdefiniowanego układu 
współrzędnych użytkownika z prędkością 
33 cm/min dopóki na wejście 1 nie będzie 
podawana wartości 1. Sygnał ten 
wysyłany jest z lasera pomiarowego. Jeśli 
laser jest poza zakresem pomiarowym 
podawane jest 0, a jeśli mieści się w 
zakresie na wejściu podawana jest 1. Po 
pojawieniu się 1 na wejściu pierwszym 
następuje zatrzymanie ruchu robota. 
Parametr PL=0 (Position Level) określa 



aby podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

GETS PX001 $PX001 ; Pobranie zmiennej systemowej 
opisującej punkt jako współrzędną, czyli 
przechwyt współrzędnych punktu 
zatrzymania i zapis w zmiennej 
pozycyjnej P001.

GETE D001 P001 (1) ; Pobranie wartości współrzędnej X z 
zapisanej zmiennej pozycyjnej P001 i 
wstawienie jej do zmiennej D001 typu 
DOUBLE.

MOVL V=33 PL=0 UNTIL IN#(1)=OFF ; Ruch z interpolacją liniową do początku 
zdefiniowanego układu współrzędnych 
użytkownika z prędkością 33 cm/min 
dopóki na wejście 1 nie będzie podawana 
wartości 0. Sygnał ten wysyłany jest z 
lasera pomiarowego. Jeśli laser jest poza 
zakresem pomiarowym podawane jest 0, a
jeśli mieści się w zakresie na wejściu 
podawana jest 1. Po pojawieniu się 0 na 
wejściu pierwszym następuje zatrzymanie 
ruchu robota. Parametr PL=0 (Position 
Level) określa aby podczas ruchu robot 
dokładnie osiągnął zaprogramowaną 
pozycję.

GETS PX002 $PX001 ; Pobranie zmiennej systemowej 
opisującej punkt jako współrzędną, czyli 
przechwyt współrzędnych punktu 
zatrzymania i zapis w zmiennej 
pozycyjnej P002.

GETE D002 P002 (1) ; Pobranie wartości współrzędnej X z 
zapisanej zmiennej pozycyjnej P002 i 
wstawienie jej do zmiennej D002 typu 
DOUBLE.

SET D003 (D001 + D002)/2 ; Obliczenie wartości współrzędnej X 
środka elementu i zapisanie jej do 
zmiennej D003 typu DOUBLE.

SET P003 P001 ; Wstawienie wartości zmiennej 
pozycyjnej P001 do zmiennej pozycyjnej 
P003, a więc przyjęcie punktu 3 jako 
punkt 1.



SETE P003 (1) D003 ; Wstawienie wartości zmiennej D003, 
czyli współrzędnej środka elementu w osi 
X, do zmiennej pozycyjnej P003 na 
współrzędną pierwszą (x).

MOVL P003 V=138 PL=0 ; Ruch z interpolacją liniową do 
wyznaczonego środka elementu w osi X z 
prędkością 138 cm/min. Parametr PL=0 
(Position Level) określa aby podczas 
ruchu robot dokładnie osiągnął 
zaprogramowaną pozycję.

SET P004 P003 ; Wstawienie wartości zmiennej 
pozycyjnej P003 do zmiennej pozycyjnej 
P004, a więc przyjęcie punktu 4 jako 
punkt 3.

GETE D004 P003 (2) ; Pobranie wartości współrzędnej Y z 
zapisanej zmiennej pozycyjnej P003 i 
wstawienie jej do zmiennej D004 typu 
DOUBLE.

SET D005 D004 - 100000 ; Zapisanie do zmiennej D005 typu 
DOUBLE wartości zmiennej D004 
pomniejszonej o 100000μm.

SETE P004 (2) D005 ; Wstawienie wartości zmiennej D005 do 
zmiennej pozycyjnej P004 na współrzędną
drugą (y).

MOVL P004 V=33 PL=0 UNTIL IN#(1)=OFF ; Ruch z interpolacją liniową do 
wyznaczonego punktu P004 z 
prędkością 33 cm/min dopóki na 
wejście 1 nie będzie podawana 
wartości 0. Sygnał ten wysyłany 
jest z lasera pomiarowego. Jeśli 
laser jest poza zakresem 
pomiarowym podawane jest 0, a 
jeśli mieści się w zakresie na 
wejściu podawana jest 1. Po 
pojawieniu się 0 na wejściu 
pierwszym następuje zatrzymanie 
ruchu robota. Parametr PL=0 
(Position Level) określa aby 
podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

GETS PX005 $PX001 ; Pobranie zmiennej systemowej 
opisującej punkt jako współrzędną, czyli 
przechwyt współrzędnych punktu 



zatrzymania i zapis w zmiennej 
pozycyjnej P005.

GETE D005 P005 (2) ; Pobranie wartości współrzędnej Y z 
zapisanej zmiennej pozycyjnej P005 i 
wstawienie jej do zmiennej D005 typu 
DOUBLE.

MOVL P004 V=33 PL=0 UNTIL IN#(1)=ON ; Ruch z interpolacją liniową do 
punktu P004 z prędkością 33 
cm/min dopóki na wejście 1 nie 
będzie podawana wartości 1. 
Sygnał ten wysyłany jest z lasera 
pomiarowego. Jeśli laser jest poza 
zakresem pomiarowym podawane 
jest 0, a jeśli mieści się w zakresie 
na wejściu podawana jest 1. Po 
pojawieniu się 1 na wejściu 
pierwszym następuje zatrzymanie 
ruchu robota. Parametr PL=0 
(Position Level) określa aby 
podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

GETS PX006 $PX001 ; Pobranie zmiennej systemowej 
opisującej punkt jako współrzędną, czyli 
przechwyt współrzędnych punktu 
zatrzymania i zapis w zmiennej 
pozycyjnej P006.

GETE D006 P006 (2) ; Pobranie wartości współrzędnej Y z 
zapisanej zmiennej pozycyjnej P006 i 
wstawienie jej do zmiennej D006 typu 
DOUBLE.

SET D007 (D005 + D006)/2 ; Obliczenie wartości współrzędnej Y 
środka elementu i zapisanie jej do 
zmiennej D007 typu DOUBLE.

SET P007 P006 ; Wstawienie wartości zmiennej 
pozycyjnej P006 do zmiennej pozycyjnej 
P007, a więc przyjęcie punktu 7 jako 
punkt 6.

SETE P007 (2) D007 ; Wstawienie wartości zmiennej D007, 
czyli współrzędnej środka elementu w osi 
Y, do zmiennej pozycyjnej P007 na 
współrzędną drugą (y).



MOVL P007 V=33 PL=0 ; Ruch z interpolacją liniową do 
wyznaczonego środka elementu w osi Y z 
prędkością 33 cm/min. Parametr PL=0 
(Position Level) określa aby podczas 
ruchu robot dokładnie osiągnął 
zaprogramowaną pozycję.

GETE D008 P003 (2) ; Pobranie wartości współrzędnej Y z 
zapisanej zmiennej pozycyjnej P003 i 
wstawienie jej do zmiennej D008 typu 
DOUBLE.

SET R001 (D002 - D003)/1000 ; Obliczenie długości jednej z 
przyprostokątnych utworzonego trójkąta 
prostokątnego i zapisanie tej długości w 
zmiennej R001 typu rzeczywistego.

SET R002 (D007 - D008)/1000 ; Obliczenie długości drugiej 
przyprostokątnej utworzonego trójkąta 
prostokątnego i zapisanie tej długości w 
zmiennej R002 typu rzeczywistego.

SET R003 R001 * R001 + R002 * R002 ; Obliczenie sumy kwadratów długości 
przyprostokątnych i zapisanie jej w 
zmiennej R003 typu rzeczywistego. 

SET SQRT R000 R003  ; Obliczenie pierwiastka z wyznaczonej 
wcześniej sumy kwadratów, co odpowiada
długości przeciwprostokątnej, a ta jest 
równa promieniowi otworu w elemencie. 
Wartość promienia zapisujemy do 
zmiennej R000.

SET R004 R000 * 2 ; Obliczenie średnicy otworu i zapisanie 
do zmiennej R004.

Po sprawdzeniu dokładności z jaką określany jest punkt środkowy elementu przez 
robota widzimy, że błąd pozycjonowania wynosi ok. 1mm w osi X i ok. 3mm w osi Y. 
Dlatego w programie wprowadzamy korektę w pozycjonowaniu.

SET D009 D003 + 1000 ; Dodanie 1000μm do zmiennej D003 
zawierającej wartość współrzędnej X 
punktu środkowego elementu i zapisanie 
tej wartości w zmiennej D009 typu 
DOUBLE.

SET D010 D007 + 3000 ; Dodanie 3000μm do zmiennej D007 
zawierającej wartość współrzędnej Y 



punktu środkowego elementu i zapisanie 
tej wartości w zmiennej D010 typu 
DOUBLE.

SET P008 P007 ; Wstawienie wartości zmiennej 
pozycyjnej P007 do zmiennej pozycyjnej 
P008, a więc przyjęcie punktu 8 jako 
punkt 7.

SETE P008 (1) D009 ; Wstawienie wartości zmiennej D009, 
czyli wprowadzonej poprawki 
współrzędnej środka elementu w osi X, do
zmiennej pozycyjnej P008 na współrzędną
pierwszą (x).

SETE P008 (2) D010 ; Wstawienie wartości zmiennej D010, 
czyli wprowadzonej poprawki 
współrzędnej środka elementu w osi Y, do
zmiennej pozycyjnej P008 na współrzędną
drugą (y).

MOVL P008 V=100 PL=0 ; Ruch z interpolacją liniową do środka 
elementu po wprowadzonych korektach z 
prędkością 100 cm/min. Parametr PL=0 
(Position Level) określa aby podczas 
ruchu robot dokładnie osiągnął 
zaprogramowaną pozycję.

GETE D011 P008 (3) ; Pobranie wartości współrzędnej Z z 
zapisanej zmiennej pozycyjnej P008 i 
wstawienie jej do zmiennej D011 typu 
DOUBLE.

SET D012 D011 - 90000 ; Obliczenie współrzędnej Z, która będzie 
po obniżeniu chwytaka w celu złapania 
elementu i zapisanie jej wartości w 
zmiennej D012 typu DOUBLE.

SET P009 P008 ; Wstawienie wartości zmiennej 
pozycyjnej P008 do zmiennej pozycyjnej 
P009, a więc przyjęcie punktu 9 jako 
punkt 8.

SETE P009 (3) D012 ; Wstawienie wartości zmiennej D012 do 
zmiennej pozycyjnej P009 na współrzędną
trzecią (z).



MOVL P009 V=100 PL=0 ; Ruch z interpolacją liniową do punktu 
P009, czyli do otworu znajdującego się w 
elemencie z prędkością 100 cm/min. 
Parametr PL=0 (Position Level) określa 
aby podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

DOUT OT#(4) OFF ; Ustawienie wartości 0 na wyjściu 
dyskretnym 4 przypisanym do chwytaka. 
Powoduje to otwarcie chwytaka i złapanie 
przedmiotu.

MOVL P008 V=100 PL=0 ; Ruch z interpolacją liniową z powrotem 
do punktu P008, czyli uniesienie elementu
w górę z prędkością 100 cm/min. 
Parametr PL=0 (Position Level) określa 
aby podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

SET D013 D010 + 100000 ; Dodanie 100000μm do zmiennej D010, 
czyli współrzędnej Y środka elementu i 
zapisanie tej wartości w zmiennej D013 
typu DOUBLE.

SET D014 D009 + 100000 ; Dodanie 100000μm do zmiennej D009, 
czyli współrzędnej X środka elementu i 
zapisanie tej wartości w zmiennej D014 
typu DOUBLE.

SET P010 P008 ; Wstawienie wartości zmiennej 
pozycyjnej P008 do zmiennej pozycyjnej 
P010, a więc przyjęcie punktu 10 jako 
punkt 8.

SETE P010 (1) D014 ; Wstawienie wartości zmiennej D014, 
czyli współrzędnej X odłożenia elementu, 
do zmiennej pozycyjnej P010 na 
współrzędną pierwszą (x).

SETE P010 (2) D013 ; Wstawienie wartości zmiennej D013, 
czyli współrzędnej Y odłożenia elementu, 
do zmiennej pozycyjnej P010 na 
współrzędną drugą (y).

MOVL P010 V=138 PL=0 ; Ruch z interpolacją liniową do punktu 
P010, czyli do punktu odłożenia elementu 
z prędkością 138 cm/min. Parametr PL=0 
(Position Level) określa aby podczas 



ruchu robot dokładnie osiągnął 
zaprogramowaną pozycję.

GETE D015 P010 (3) ; Pobranie wartości współrzędnej Z z 
zapisanej zmiennej pozycyjnej P010 i 
wstawienie jej do zmiennej D015 typu 
DOUBLE.

SET D016 D015 - 90000 ; Obliczenie współrzędnej Z, która będzie 
po obniżeniu chwytaka w celu odłożenia 
elementu i zapisanie jej wartości w 
zmiennej D016 typu DOUBLE.

SET P011 P010 ; Wstawienie wartości zmiennej 
pozycyjnej P010 do zmiennej pozycyjnej 
P011, a więc przyjęcie punktu 11 jako 
punkt 10.

SETE P011 (3) D016 ; Wstawienie wartości zmiennej D016 do 
zmiennej pozycyjnej P011 na współrzędną
trzecią (z).

MOVL P011 V=138 PL=0 ; Ruch z interpolacją liniową do punktu 
P011, czyli w dół do punktu odłożenia 
elementu z prędkością 138 cm/min. 
Parametr PL=0 (Position Level) określa 
aby podczas ruchu robot dokładnie 
osiągnął zaprogramowaną pozycję.

DOUT OT#(4) ON ; Ustawienie wartości 1 na wyjściu 
dyskretnym 4 przypisanym do chwytaka. 
Powoduje to zamknięcie chwytaka i 
odłożenie przedmiotu.

MOVL P010 V=138 PL=0 ; Ruch z interpolacją liniową do punktu 
P010, czyli z powrotem do góry, nad 
odłożony element z prędkością 138 
cm/min. Parametr PL=0 (Position Level) 
określa aby podczas ruchu robot dokładnie
osiągnął zaprogramowaną pozycję.

END ; Koniec programu.

W załączniku do instrukcji wykonywania ćwiczenia zamieszczono kody źródłowe 
programów do określania średnicy elementu oraz sortowania.



Rys. 10. Stanowisko doświadczalne robota przemysłowego YASKAWA
MOTOMAN UP6.

Rys. 11. Przenoszenie elementów z wykorzystaniem robota przemysłowego
YASKAWA MOTOMAN UP6.



Rys.12. Ekran panelu operatorskiego i wybór funkcji.

Załącznik 1- program główny:

/JOB
//NAME SORT1
//POS
///NPOS 6,0,0,12,0,0
///TOOL 0
///POSTYPE PULSE
///PULSE
C00000=92645,5038,7909,1221,-1859,26520
C00001=159556,-7882,-4797,792,-2710,8648
C00002=159554,-7882,-4797,793,-2709,8650
C00003=159556,-7882,-4797,792,-2710,8648
///TOOL 2
C00004=159556,-7883,-4797,793,-2710,8647
///TOOL 0
C00005=92643,5038,7909,1222,-1859,26522
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0001=237.689,707.258,627.566,179.54,1.30,123.32
///TOOL 2



P0002=174.740,707.259,627.562,179.54,1.30,123.32
///TOOL 0
P0003=206.214,707.258,627.566,179.54,1.30,123.32
P0004=206.214,607.258,627.566,179.54,1.30,123.32
P0005=206.214,673.133,627.566,179.54,1.30,123.32
P0006=206.212,635.843,627.567,179.54,1.30,123.32
P0007=206.212,654.488,627.567,179.54,1.30,123.32
P0008=207.214,657.488,627.567,179.54,1.30,123.32
P0009=207.214,657.488,527.567,179.54,1.30,123.32
P0012=418.290,707.257,627.566,179.54,1.30,123.31
P0030=243.985,707.258,627.566,179.54,1.30,123.32
P0031=183.055,707.259,627.562,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:08
///ATTR SC,RW
///GROUP1 RB1
NOP
DOUT OT#(4) ON
MOVL C00000 V=33.3
*SKAN
GETS PX012 $PX001
GETE D017 P012 (1)
GETE D018 P012 (2)
MOVL C00001 V=33.3 PL=0 UNTIL IN#(1)=ON
GETS PX001 $PX001
GETE D001 P001 (1)
GETS PX030 $PX001
GETE D030 P001 (1)
SET D030  EXPRESS D001 + 15000
SETE P030 (1) D030
MOVL P030 V=33.3 PL=0
MOVL C00002 V=3.3 PL=0 UNTIL IN#(1)=ON
GETS PX001 $PX001
GETE D001 P001 (1)
JUMP *KONIEC IF D001<-154000
MOVL C00003 V=33.3 PL=0 UNTIL IN#(1)=OFF
GETS PX002 $PX001
GETE D002 P002 (1)
GETS PX031 $PX001
GETE D031 P002 (1)
SET D031  EXPRESS D002 + 16000
SETE P031 (1) D031
MOVL P031 V=33.3 PL=0
MOVL C00004 V=3.3 PL=0 UNTIL IN#(1)=OFF
GETS PX002 $PX001
GETE D002 P002 (1)
SET D003  EXPRESS ( D001 + D002 ) / 2
SET P003 P001
SETE P003 (1) D003
MOVL P003 V=33.3 PL=0



SET P004 P003
GETE D004 P003 (2)
SET D005  EXPRESS D004 - 100000
SETE P004 (2) D005
MOVL P004 V=5.5 PL=0 UNTIL IN#(1)=OFF
GETS PX005 $PX001
GETE D005 P005 (2)
MOVL P004 V=5.5 PL=0 UNTIL IN#(1)=ON
GETS PX006 $PX001
GETE D006 P006 (2)
SET D007  EXPRESS ( D005 + D006 ) / 2
SET P007 P006
SETE P007 (2) D007
MOVL P007 V=33.3 PL=0
GETE D008 P003 (2)
SET R001  EXPRESS ( D002 - D003 ) / 1000
SET R002  EXPRESS ( D007 - D008 ) / 1000
SET R003  EXPRESS R001 * R001 + R002 * R002
SQRT R000 R003
SET R004  EXPRESS R000 * 2
SET D009  EXPRESS D003 + 1000
SET D010  EXPRESS D007 + 3000
SET P008 P007
SETE P008 (1) D009
SETE P008 (2) D010
MOVL P008 V=11.0 PL=0
GETE D011 P008 (3)
SET D012  EXPRESS D011 - 100000
SET P009 P008
SETE P009 (3) D012
MOVL P009 V=11.0 PL=0
TIMER T=2.00
DOUT OT#(4) OFF
MOVL P008 V=16.7 PL=0
TIMER T=2.00
SET R005  EXPRESS ( R004 - 72 ) * ( R004 - 72 )
SET R006  EXPRESS ( R004 - 82 ) * ( R004 - 82 )
SET R007  EXPRESS ( R004 - 92 ) * ( R004 - 92 )
SET R008  EXPRESS ( R004 - 102 ) * ( R004 - 102 )
SET R009  EXPRESS ( R004 - 112 ) * ( R004 - 112 )
SET R010  EXPRESS ( R004 - 122 ) * ( R004 - 122 )
CALL JOB:PRZEN IF R005<2
CALL JOB:PRZEN1 IF R006<2
CALL JOB:PRZEN2 IF R007<2
CALL JOB:PRZEN3 IF R008<2
CALL JOB:PRZEN4 IF R009<2
CALL JOB:PRZEN5 IF R010<2
TIMER T=2.00
MOVL C00005 V=33.3
JUMP *SKAN IF R004<115



*KONIEC
END

Załącznik 2 – programy na sortowanie elementu:
/JOB
//NAME PRZEN
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=207.214,657.488,627.567,179.54,1.30,123.32
P0010=718.290,627.257,627.567,179.54,1.30,123.32
P0011=718.290,627.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:12
///ATTR SC,RW
///GROUP1 RB1
NOP
SET D013  EXPRESS D017 - 150000
SET D014  EXPRESS D018 + 150000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0
END

/JOB
//NAME PRZEN1
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=267.385,683.353,627.567,179.54,1.30,123.32
P0010=318.290,857.257,627.567,179.54,1.30,123.32
P0011=318.290,857.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/10 10:35



///ATTR SC,RW
///GROUP1 RB1
NOP
SET D013  EXPRESS D017 - 200000
SET D014  EXPRESS D018 + 150000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0
END

/JOB
//NAME PRZEN2
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=267.385,683.353,627.567,179.54,1.30,123.32
P0010=318.290,857.257,627.567,179.54,1.30,123.32
P0011=318.290,857.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:12
///ATTR SC,RW
///GROUP1 RB1
NOP
SET D013  EXPRESS D017 - 330000
SET D014  EXPRESS D018 + 150000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0



END

/JOB
//NAME PRZEN3
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=267.385,683.353,627.567,179.54,1.30,123.32
P0010=318.290,857.257,627.567,179.54,1.30,123.32
P0011=318.290,857.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:13
///ATTR SC,RW
///GROUP1 RB1
NOP
SET D013  EXPRESS D017 + 150000
SET D014  EXPRESS D018 + 150000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0
END

/JOB
//NAME PRZEN4
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=267.385,683.353,627.567,179.54,1.30,123.32
P0010=318.290,857.257,627.567,179.54,1.30,123.32
P0011=318.290,857.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:11
///ATTR SC,RW



///GROUP1 RB1
NOP
SET D013  EXPRESS D017 - 150000
SET D014  EXPRESS D018 + 400000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0
END

/JOB
//NAME PRZEN5
//POS
///NPOS 0,0,0,3,0,0
///TOOL 0
///POSTYPE BASE
///RECTAN
///RCONF 0,0,0,0,0,0,0,0
P0008=207.214,657.488,627.567,179.54,1.30,123.32
P0010=718.290,627.257,627.567,179.54,1.30,123.32
P0011=718.290,627.257,537.567,179.54,1.30,123.32
//INST
///DATE 2013/09/06 12:11
///ATTR SC,RW
///GROUP1 RB1
NOP
SET D013  EXPRESS D017 + 300000
SET D014  EXPRESS D018 - 80000
SET P010 P008
SETE P010 (1) D013
SETE P010 (2) D014
MOVL P010 V=33.3 PL=0
GETE D015 P010 (3)
SET D016  EXPRESS D015 - 90000
SET P011 P010
SETE P011 (3) D016
MOVL P011 V=5.5 PL=0
TIMER T=2.00
DOUT OT#(4) ON
MOVL P010 V=33.3 PL=0
END 

